Ricerca Operativa - Laboratorio

Lezione 3 - Esempi di modelli PL

Docente: Luigi De Giovanni

Dipartimento di Matematica “Tullio Levi-Civita”
Università degli Studi di Padova

luigi@math.unipd.it
https://www.math.unipd.it/∼luigi/

Corso di Laurea Magistrale in Matematica
Università degli Studi di Padova
a.a. 2019–2020

Esempio 3.1

Un'azienda produce fibra di vetro al metro cubo e desidera pianificare la produzione per le prossime sei settimane. La capacità produttiva è limitata e tale limite varia al variare del periodo considerato. La domanda settimanale è nota per l'intero periodo. I costi unitari di produzione ed immagazzinamento variano al variare del periodo. I dati sono riportati nella seguente tabella.

Settimana	Capacità	Domanda	Costo Produzione	Costo Immagazzinam.
1	140	100	5	0.2
2	100	120	8	0.3
3	110	100	6	0.2
4	100	90	6	0.25
5	120	120	7	0.3
6	100	110	6	0.4

Pianificare la produzione in modo da minimizzare il costo totale di produzione e stoccaggio in due ipotesi: (i) la produzione inizia la prima settimana con il magazzino vuoto e termina la sesta settimana con il magazzino vuoto, oppure (ii) la produzione è ciclica

Problemi di pianificazione della produzione (i)

Si assuma di avere

- $\mathcal{T}$: insieme di n periodi.

Problemi di pianificazione della produzione (i)

Si assuma di avere

- $\mathcal{T}$: insieme di n periodi.

Siano

- c_prod_j: costo di produzione periodo $j \in \mathcal{T}$,
- c_imm_j: costo di immagazzinamento periodo $j \in \mathcal{T}$,
- d_j: domanda periodo $j \in \mathcal{T}$,
- cap_j: capacità produttiva periodo $j \in \mathcal{T}$,
- x_j: quantità di prodotto realizzato periodo $j \in \mathcal{T}$ (incognite),
- s_j: quantità di prodotto immagazzinato periodo $j \in \mathcal{T}$ (incognite).

Problemi di pianificazione della produzione (i)

Si assuma di avere

- $\mathcal{T}$: insieme di n periodi.

Siano

- c_prod_j: costo di produzione periodo $j \in \mathcal{T}$,
- c_imm_j: costo di immagazzinamento periodo $j \in \mathcal{T}$,
- d_j: domanda periodo $j \in \mathcal{T}$,
- cap_j: capacità produttiva periodo $j \in \mathcal{T}$,
- x_j: quantità di prodotto realizzato periodo $j \in \mathcal{T}$ (incognite),
- s_j: quantità di prodotto immagazzinato periodo $j \in \mathcal{T}$ (incognite).

Obiettivo: pianificare la produzione nel periodo $\mathcal{T}$, minimizzando i costi e rispettando i vincoli di domanda e capacità di ogni periodo.

Problemi di pianificazione della produzione (ii)

$$\min \sum_{j \in \mathcal{T}} (c\_prod_j \cdot x_j) + \sum_{j \in \mathcal{T}} (c\_imm_j \cdot s_j)$$

$$x_1 + \alpha = d_1 + s_1$$

$$x_j + s_{j-1} = d_j + s_j, \quad j \in \mathcal{T}[, j \geq 2(ipotesi(ii))]$$

$$0 \leq x_j \leq cap_j, \quad j \in \mathcal{T}$$

$$s_j \geq 0, \quad j \in \mathcal{T}$$

$$s_n = \alpha$$

Sostituire α con 0 nell' ipotesi (i), con s_n nell'ipotesi (ii)

Problemi di pianificazione della produzione (iv)

pianificazione.mod

```
set T circular;

param cap{T};
param domanda{T};
param c_prod{T};
param c_imm{T};

var x{j in T} >= 0, <= cap[j];
var s{j in T} >= 0;

minimize f: sum{j in T} c_prod[j]*x[j]
            + sum{j in T} c_imm[j]*s[j];

s.t. v{j in T}: x[j] + s[prev(j,T)] = domanda[j] + s[j];
```

Problemi di pianificazione della produzione (v)

pianificazione.fibra.dat

```
set T:= sett1 sett2 sett3 sett4 sett5 sett6;

param cap :=
sett1 140
sett2 100
sett3 110
sett4 100
sett5 120
sett6 100;

param: domanda c_prod c_imm :=
sett1    100      5      0.2
sett2    120      8      0.3
sett3    100      6      0.2
sett4     90      6      0.25
sett5    120      7      0.3
sett6    110      6      0.4;
```

Problemi di pianificazione della produzione (vi)

fibra.run

```
reset;

model pianificazione.mod;
data pianificazione.fibra.dat;

fix s[last(T)] := 0;

option solver cplex;
solve;

display f;
display x;
display s;
```

Esempio 3.2

LegnoPadova SpA è un'azienda specializzata nella produzione di parquet. I suoi stabilimenti producono vari tipi di parquet che vengono poi venduti sul mercato generando un determinato profitto. Ciascun settore della produzione può lavorare un numero limitato di ore (mensilmente), come riportato nella seguente tabella.

	Falegnameria	Verniciatura	Assemblaggio	Verifica	Imballaggio
Ore	2000	1500	1700	300	500

Nella seguente tabella si riporta il profitto unitario e la domanda massima per tipologia di parquet.

	Supereconomy	Economy	Biocity	Vintage	Gaia	Family	Extralux
Profitto	10	18	20	25	27	28	35
Domanda	100	80	60	40	40	20	20

Infine, le ore di lavorazione necessarie per ciascuna tipologia di parquet sono riportate nella seguente tabella.

	Superecon.	Economy	Biocity	Vintage	Gaia	Family	Extralux
Falegnameria	7	5	9	10	10	12	15
Verniciatura	2	2	2	3	3	3	3
Assemblaggio	2	2	4	7	9	15	18
Verifica	1	1	1	2	1	2	2
Imballaggio	1	1	1	1	2	1	0

Determinare il piano produttivo che massimizza il profitto totale.

Problemi di ripartizione di risorse (i)

Si assuma di avere

- $\mathcal{P}$: insieme di n prodotti,
- $\mathcal{R}$: insieme di m risorse.

Problemi di ripartizione di risorse (i)

Si assuma di avere

- $\mathcal{P}$: insieme di n prodotti,
- $\mathcal{R}$: insieme di m risorse.

Siano

- b_i: quantità (massima) disponibile di risorsa $i \in \mathcal{R}$,
- c_j: profitto di vendita unitario del prodotto $j \in \mathcal{P}$,
- d_j: domanda massima del prodotto $j \in \mathcal{P}$,
- a_{ij}: quantità di risorsa $i \in \mathcal{R}$ necessaria per realizzare una unità di prodotto $j \in \mathcal{P}$,
- x_j: quantità di prodotto $j \in \mathcal{P}$ da realizzare (incognite del problema).

Problemi di ripartizione di risorse (i)

Si assuma di avere

- $\mathcal{P}$: insieme di n prodotti,
- $\mathcal{R}$: insieme di m risorse.

Siano

- b_i: quantità (massima) disponibile di risorsa $i \in \mathcal{R}$,
- c_j: profitto di vendita unitario del prodotto $j \in \mathcal{P}$,
- d_j: domanda massima del prodotto $j \in \mathcal{P}$,
- a_{ij}: quantità di risorsa $i \in \mathcal{R}$ necessaria per realizzare una unità di prodotto $j \in \mathcal{P}$,
- x_j: quantità di prodotto $j \in \mathcal{P}$ da realizzare (incognite del problema).

Obiettivo: massimizzare i profitti, rispettando i vincoli di domanda e di limitatezza delle risorse.

Problemi di ripartizione di risorse (ii)

$$\max \sum_{j \in \mathcal{P}} c_j x_j$$

$$\sum_{j \in \mathcal{P}} a_{ij} x_j \leq b_i, \quad i \in \mathcal{R}$$

$$0 \leq x_j \leq d_j, \quad j \in \mathcal{P}$$

Problemi di ripartizione di risorse (iv)

parquet.mod

```
set PARQUET;
set REPARTI;

param max_ore{REPARTI};
param profitto{PARQUET};
param domanda{PARQUET};
param ore_lav{REPARTI,PARQUET};

var x{j in PARQUET} >= 0, <= domanda[j];

maximize f: sum{j in PARQUET} profitto[j]*x[j];

s.t. v{i in REPARTI}:
     sum{j in PARQUET} ore_lav[i,j]*x[j] <= max_ore[i];
```

Problemi di ripartizione di risorse (v)

parquet.dat

```
set PARQUET := Supec Ec Bioc Vint Gaia Family Extralux;
set REPARTI := Faleg Vern Assem Ver Imbal;

param max_ore :=
Faleg  2000
Vern   1500
Assem  1700
Ver     300
Imbal   500;
```

Problemi di ripartizione di risorse (vi)

```
param: profitto domanda :=
Supec      10       100
Ec         18        80
Bioc       20        60
Vint       25        40
Gaia       27        40
Family     28        20
Extralux   35        20;

param ore_lav: Supec Ec Bioc Vint Gaia Family Extralux :=
Faleg              7    5    9   10   10     12       15
Vern               2    2    2    3    3      3        3
Assem              2    2    4    7    9     15       18
Ver                1    1    1    2    1      2        2
Imbal              1    1    1    1    2      1        0;
```

Problemi di ripartizione di risorse (vii)

parquet.run

```
reset;

model parquet.mod;
data parquet.dat;

option solver cplex;

solve;

display f;
display x;
```

Esercizio proposto (i)

Esercizio 3.1

L'azienda Ovile produce due tipi di cibo per animali: granulare e in polvere. Le materie prime utilizzate per la produzione sono: avena, mais e melassa. Tali materie, ad eccezione della melassa, devo essere macinate prima della lavorazione. In seguito si mescolano le varie materie e si processa il composto (granulazione o polverizzazione) al fine di ottenere i due diversi tipi di prodotto. La percentuale di sostanze nutritive (proteine, grassi e fibre) contenute nelle materie prime e i requisiti nutrizionali (in %) che i prodotti devono soddisfare sono riportati in tabella.

Materie prime	Proteine	Grassi	Fibre
Avena	13.6	7.1	7
Mais	4.1	2.4	3.7
Melassa	5	0.3	25
Requisiti	≥ 9.5	≥ 2	≤ 6

Di seguito sono riportati la disponibilità delle materie prime e i costi unitari per il loro acquisto.

Materie prime	Disponibilità (kg)	Costo (Euro/kg)
Avena	11900	0.13
Mais	23500	0.17
Melassa	750	0.12

Infine, i costi di produzione (per un kg di materie prime) sono riportati nella seguente tabella.

Macina	Mescola	Granulazione	Polverizzazione
0.25	0.05	0.42	0.17

Tenendo conto che la domanda giornaliera (esatta) è di 9 tonnellate per il prodotto granulare e di 12 tonnellate per quello in polvere, determinare il piano produttivo che minimizza il costo totale.

Esercizio proposto (ii)

Variabili: x_{ij} è la quantità (in kg) di materia prima i destinata al tipo di prodotto j

Esercizio proposto (ii)

Variabili: x_{ij} è la quantità (in kg) di materia prima i destinata al tipo di prodotto j

$$
\begin{aligned}
\min\ & 0.13(x_{11}+x_{12})+0.17(x_{21}+x_{22})+0.12(x_{31}+x_{32})\\
& +0.25(x_{11}+x_{12}+x_{21}+x_{22})\\
& +0.05(x_{11}+x_{12}+x_{21}+x_{22}+x_{31}+x_{32})\\
& +0.42(x_{11}+x_{21}+x_{31})\\
& +0.17(x_{12}+x_{22}+x_{32})
\end{aligned}
$$

$$
\begin{aligned}
& 0.136x_{11}+0.041x_{21}+0.05x_{31}\geq 0.095(x_{11}+x_{21}+x_{31})\\
& 0.071x_{11}+0.024x_{21}+0.003x_{31}\geq 0.02(x_{11}+x_{21}+x_{31})\\
& 0.07x_{11}+0.037x_{21}+0.25x_{31}\leq 0.06(x_{11}+x_{21}+x_{31})\\
& 0.136x_{12}+0.041x_{22}+0.05x_{32}\geq 0.095(x_{12}+x_{22}+x_{32})\\
& 0.071x_{12}+0.024x_{22}+0.003x_{32}\geq 0.02(x_{12}+x_{22}+x_{32})\\
& 0.07x_{12}+0.037x_{22}+0.25x_{32}\leq 0.06(x_{12}+x_{22}+x_{32})\\
& x_{11}+x_{12}\leq 11900\\
& x_{21}+x_{22}\leq 23500\\
& x_{31}+x_{32}\leq 750\\
& x_{11}+x_{21}+x_{31}=9000\\
& x_{12}+x_{22}+x_{32}=12000\\
& x_{ij}\geq 0 \quad i=1,\ldots,3 \quad j=1,2
\end{aligned}
$$

Esercizio proposto (iii)

- Insiemi: I (MATERIE); J (CIBI); K (SOSTANZE); R (LAVORAZIONI).
- Parametri: ...; $P_{rij} = 1$ se è richiesta la lavorazione r sulla materia i per il cibo j, 0 altrimenti; ...

Esercizio proposto (iii)

- Insiemi: I (MATERIE); J (CIBI); K (SOSTANZE); R (LAVORAZIONI).
- Parametri: ...; P_{rij} = 1 se è richiesta la lavorazione r sulla materia i per il cibo j, 0 altrimenti; ...
- Modello PL:

$$
\begin{array}{lll}
\min & \sum_{i \in I} C_i \sum_{j \in J} x_{ij} + \sum_{r \in R} F_r \sum_{i \in I} \sum_{j \in J} P_{rij} x_{ij} & \\
s.t. & & \\
& \sum_{i \in I} A_{ik} x_{ij} \geq B_k \sum_{i \in I} x_{ij} & \forall\, j \in J, k \in K : B_{kj} > 0 \\
& \sum_{i \in I} A_{ik} x_{ij} \leq U_k \sum_{i \in I} x_{ij} & \forall\, j \in J, k \in K : U_{kj} < 1 \\
& \sum_{j \in J} x_{ij} \leq Q_i & \forall\, i \in I, \\
& \sum_{i \in I} x_{ij} = D_j & \forall\, j \in J, \\
& x_{ij} \in \mathbb{R}_+ & \forall\, i \in I, j \in J
\end{array}
$$

Esercizio proposto (iv)

ovile.mod

```
set MATERIE;
set CIBI;
set SOSTANZE;
set LAVORI;

param costi_lavorazione{LAVORI};
param costi_acquisto{MATERIE};
param apporto_unitario{MATERIE,SOSTANZE} >= 0, <= 1;
param perc_min{SOSTANZE} >= 0, <= 1;
param perc_max{SOSTANZE} >= 0, <= 1;
param disp{MATERIE};
param domanda{CIBI};
param lavori_bool{LAVORI,MATERIE,CIBI} binary;

var x{MATERIE,CIBI} >= 0;
```

Esercizio proposto (v)

```
minimize f :
  sum{i in MATERIE, j in CIBI} costi_acquisto[i]*x[i,j]
  + sum{r in LAVORI, i in MATERIE, j in CIBI}
    lavori_bool[r,i,j]*costi_lavorazione[r]*x[i,j];

s.t. v_perc_min{j in CIBI, k in SOSTANZE} :
  sum{i in MATERIE} apporto_unitario[i,k]*x[i,j] >=
  perc_min[k]*sum{i in MATERIE} x[i,j];
s.t. v_perc_max{j in CIBI, k in SOSTANZE} :
  sum{i in MATERIE} apporto_unitario[i,k]*x[i,j] <=
  perc_max[k]*sum{i in MATERIE} x[i,j];
s.t. v_disp{i in MATERIE} :
  sum{j in CIBI} x[i,j] <= disp[i];
s.t. v_domanda{j in CIBI} :
  sum{i in MATERIE} x[i,j] = domanda[j];
```

Esercizio proposto (vi)

ovile.dat

```
set CIBI := Granulare Polvere;
set MATERIE := Avena Mais Melassa;
set SOSTANZE := Proteine Grassi Fibre;
set LAVORI := Macina Mescola Granulazione Polverizzazione;

param costi_lavorazione :=
Macina          0.25
Mescola         0.05
Granulazione    0.42
Polverizzazione 0.17;

param : costi_acquisto  disp :=
Avena       0.13        11900
Mais        0.17        23500
Melassa     0.12          750;
```

Esercizio proposto (vii)

```
param apporto_unitario : Proteine Grassi Fibre :=
Avena                   0.136    0.071   0.07
Mais                    0.041    0.024   0.037
Melassa                 0.05     0.003   0.25;

param perc_min :=
Proteine 0.095
Grassi   0.02
Fibre    0;

param perc_max :=
Proteine 1
Grassi   1
Fibre    0.06;

param domanda :=
Granulare  9000
Polvere   12000;
```

Esercizio proposto (viii)

```
param lavori_bool :=
  [*,*,Granulare] : Avena Mais Melassa :=
  Macina                1     1     0
  Mescola               1     1     1
  Granulazione          1     1     1
  Polverizzazione       0     0     0
  [*,*,Polvere]   : Avena Mais Melassa :=
  Macina                1     1     0
  Mescola               1     1     1
  Granulazione          0     0     0
  Polverizzazione       1     1     1;
```

Esercizio proposto (ix)

ovile.run

```
reset;
model ovile.mod;
data ovile.dat;

option solver cplex;
solve;

display f;
display x;
```